

Calcul intégral (2)

Jean-Michel Sarlat

10 avril 2003

Ce document a été généré par la **fabrique de Syracuse** à partir d'un fichier texte contenant des spécifications de contenu et des commandes d'interfaçage avec **Maxima**. Le traitement de ce fichier a donné lieu, dans le même temps que ce document, à la création d'une page html.

URL : <http://melusine.eu.org/syracuse/maxima/>

La nouveauté par rapport à la précédente *feuille de calcul* portant sur les intégrales est le filtrage des sorties de **Maxima** de façon à corriger les notations des fonctions usuelles. Pour le reste, l'apprentissage de la mise en œuvre des simplifications continue.

1 Énoncé

Calculer une primitive des fonctions suivantes :

1/ $x \mapsto \frac{1}{x^2 + a^2}$

2/ $x \mapsto \frac{1}{x^2 - a^2}$

3/ $x \mapsto \frac{1}{\sqrt{x^2 + a^2}}$

4/ $x \mapsto \frac{\sqrt{x^2 - a^2}}{x}$

5/ $x \mapsto \frac{1}{\sin^2 x}$

6/ $x \mapsto \frac{1}{\sin x}$

7/ $x \mapsto \frac{1}{\cos x}$

8/ $x \mapsto \frac{\sin^3 x}{\sqrt{\cos x}}$

a désigne un réel strictement positif.

2 Calculs

```
(C2) load("integration.mc")$
```

```
(C3) assume(a>0)$
```

```
(C4) primitive(1/(x^2+a^2),x);
```

(D4)
$$\int \frac{1}{x^2 + a^2} dx = \frac{\text{Arctan}\left(\frac{x}{a}\right)}{a}$$

(C5) primitive(1/(x^2-a^2),x);

$$(D5) \quad \int \frac{1}{x^2 - a^2} dx = \frac{\ln(x - a)}{2a} - \frac{\ln(x + a)}{2a}$$

(C6) primitive(1/sqrt(x^2+a^2),x);

$$(D6) \quad \int \frac{1}{\sqrt{x^2 + a^2}} dx = \operatorname{Argsh}\left(\frac{x}{a}\right)$$

(C7) primitive(sqrt(x^2-a^2)/x,x);

$$(D7) \quad \int \frac{\sqrt{x^2 - a^2}}{x} dx = a \operatorname{Arcsin}\left(\frac{a}{|x|}\right) + \sqrt{x^2 - a^2}$$

(C8) primitive(1/sin(x)^2,x);

$$(D8) \quad \int \frac{1}{\sin^2 x} dx = -\frac{1}{\tan x}$$

(C9) primitive_simplifie(1/sin(x),x, strig3, radcan);

$$(D9) \quad \int \frac{1}{\sin x} dx = \ln \left| \tan\left(\frac{x}{2}\right) \right|$$

(C10) primitive_simplifie(1/cos(x),x, strig3, radcan);

$$(D10) \quad \int \frac{1}{\cos x} dx = \ln \left| \tan\left(\frac{2x + \pi}{4}\right) \right|$$

(C11) expand(primitive(sin(x)^3/sqrt(cos(x)),x));

$$(D11) \quad \int \frac{\sin^3 x}{\sqrt{\cos x}} dx = \frac{2(\cos x)^{\frac{5}{2}}}{5} - 2\sqrt{\cos x}$$

3 Macros : integration.mc

```
/* integration.mc
   Quelques macros pour présenter des calculs d'intégrales.
   JM Sarlat - 2003 - http://melusine.eu.org/syracuse/maxima/
*/

/* Présentation d'une intégrale définie */
integre(f,v,a,b) := 'integrate(f,v,a,b) = integrate(f,v,a,b);

/* Intégration en appliquant la relation de Chasles */
integre_chasles(f,v,a,b,c) :=
    'integrate(f,v,a,c) = integrate(f,v,a,b) + integrate(f,v,b,c);

/* Intégration en utilisant un calcul de limite aux bornes */
integre_limite(f,v,a,b) := 'integrate(f,v,a,b) = ldefint(f,v,a,b);

/* Intégration à l'aide d'une relation de Chasles avec calcul de limite
   aux bornes */
integre_chasles_limite(f,v,a,b,c) :=
    'integrate(f,v,a,c) = ldefint(f,v,a,b) + ldefint(f,v,b,c);

/* Calcul de primitive */
primitive(f,v) := 'integrate(f,v) = integrate(f,v);

/* Macro de simplification d'une expression trigo (1) */
strig1(e,v) := ev(e,
    cos(v)+1=sin(v)/tan(v/2),
    tan(v/2)^2+1=2*tan(v/2)/sin(v),
    tan(v/2)^2-1=2*tan(v/2)/tan(v));

/* Macro de simplification d'une expression trigo (2) */
strig2(e,v) := block([i],i:trigreduce(e),ev(i,
    2*cos(v)+2=tan(v/2)^2*(2-2*cos(v)),
    cos(2*v)=2*cos(v)^2-1));

/* Macro de simplification d'une expression trigo (3) (9 avril 2003) */
strig3(e,v) := ev(e,
    log(cos(v)-1) = 2*log(abs(tan(v/2))) + log(cos(v)+1),
    log(sin(v)-1) = -2*log(abs(tan(v/2+%pi/4))) + log(sin(v)+1));

/* Mise en oeuvre de simplifications avec procédure finale */
primitive_simplifie(f,v,env,final) :=
    'integrate(f,v) = final(env(integrate(f,v),v));

/* Changement de variable */
primitive_cv(f,v,eq,t) :=
    'integrate(f,v) = changevar('integrate(f,v),eq,t,v);
```